QT       += core gui

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

CONFIG += c++11

# The following define makes your compiler emit warnings if you use
# any Qt feature that has been marked deprecated (the exact warnings
# depend on your compiler). Please consult the documentation of the
# deprecated API in order to know how to port your code away from it.
DEFINES += QT_DEPRECATED_WARNINGS

# You can also make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
# You can also select to disable deprecated APIs only up to a certain version of Qt.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0

include (../../../public_src/httpframework/httpframework.pri)
include (../../../public_src/qglobalconfig/qglobalconfig.pri)
include (../../../public_src/qdirutils/qdirutils.pri)
include (../../../public_src/qfilestorageserver/qfilestorageserver.pri)

SOURCES += \
    main.cpp \
    mainservice.cpp \
    servicecenter.cpp \
    widget.cpp

HEADERS += \
    mainservice.h \
    servicecenter.h \
    widget.h

FORMS += \
    widget.ui

# Default rules for deployment.
qnx: target.path = /tmp/$${TARGET}/bin
else: unix:!android: target.path = /opt/$${TARGET}/bin
!isEmpty(target.path): INSTALLS += target

#程序以管理员身份运行
QMAKE_LFLAGS += /MANIFESTUAC:\"level=\'requireAdministrator\' uiAccess=\'false\'\"

# ==========================构建输出配置==============================
#配置文件指定字符集：
win32-msvc* {# MSVC
    QMAKE_CXXFLAGS += /utf-8
}
unix { #GCC/Clang等
    QMAKE_CXXFLAGS += -finput-charset=UTF-8
}

# 定义基础目标名称
CONFIG(release, debug|release){
    BASE_TARGET = FileStorageService
} else {
    BASE_TARGET = FileStorageServiced
}

# 根据架构设置不同的目标文件名
win32 {# Windows平台
    contains(QT_ARCH, x86_64) {# 64位架构
        TARGET = $$BASE_TARGET-64
    } else {# 32位架构
        TARGET = $$BASE_TARGET-32
    }
} else { # 其他平台（如Linux、macOS）
    contains(QT_ARCH, x86_64) { # 64位架构
        TARGET = $$BASE_TARGET-64
    } else:contains(QT_ARCH, i386) {# 32位架构
        TARGET = $$BASE_TARGET-32
    }
}
#构建输出路径
DESTDIR = $$PWD/../../build/$$TARGET
#构建临时文件
OBJECTS_DIR = $$PWD/../../build/makeout/$$TARGET/$$/obj
MOC_DIR = $$PWD/../../build/makeout/$$TARGET/moc
UI_DIR = $$PWD/../../build/makeout/$$TARGET/ui
RCC_DIR = $$PWD/../../build/makeout/$$TARGET/rcc

#程序信息
VERSION = 1.3.0
QMAKE_TARGET_PRODUCT = "FileStorageService"
QMAKE_TARGET_COMPANY = "ZhongKe Starlight Information Technology Co.,LTD @ YanLei"
QMAKE_TARGET_DESCRIPTION = "FileStorageService"
QMAKE_TARGET_COPYRIGHT = "ZhongKe Starlight Information Technology Co.,LTD @ YanLei"
RC_LANG = 0x0004 #语言
RC_FILE = resources.rc
